
D-Spec D-Lights

 

Agenda

A couple of reminders on basics
Mapping Indicators

Using the INDDS keyword
Overlaying the *IN array

D-spec discoveries
Lesser known capablities

Sorting and searching

External Names
The system’s 10 character limit still applies
RPG is considering adding the ability to use the ALIAS name 

But as of V5R4 it hasn’t happened

More on mapping file fields to Data Structures later

Internal Names
Up to 4096 characters long

Yes really!

Ellipsis (...) are used to continue names that require multiple lines
Not often used except for when a procedure name exceeds 14 characters
Or when the field name is indented by a number of characters

When used the balance of the definition is placed on the next line

 D ThisIsAReallyLongProcedureName...
 D                 PR

Names on the D-spec

Defining Named Constants

Specified by C in DS column
No length allowed
Value specified in the Keyword area

CONST( ..... ) is optional
IBM’s examples include it - but why bother !

Literals can be continued in keyword area of next line
Character literals require continuation characters

+ (plus) continues at first non-blank
- (minus) continues at first position 

Numeric literals do not need a continuation character 

 

DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords++++++++++++++++++++.....
 

D Upper           C                   CONST(’ABCDEFGHI-
D                                     JKLMNOPQRSTUVWXYZ’)

D Lower           C                   ’abcdefghijklmnopqrs+
D                                           tuvwxyz’

D BigNumber       C                   CONST(23456
D                                       789)

Named constants are not new except in the way they are defined.  However, you will find that due to 
the use of the new free-format Factor 2 C spec, named constants are not needed as much as they 
were in RPG/400.
 
Notice continuation of values.

Character values continued with - continue with first column in keywords on the continuation line.
Character values continued with + continue with first non-blank column in keywords on continuation 
line.
Numeric values require no continuation character and continue with the first numeric character on 
continuation line.

D-Spec Keywords - OVERLAY

OVERLAY replaces the need for From/To definitions
Note the use of *Next, which was added in V4R4

The field follows the previous one that was OVERLAY’d on the same field
This could have been used in place of the "16" in the first example

Note indentation of subfield names
Note particularly the use of a space following the D

We strongly recommend that you follow this practice

Can be used against a field name OR the DS name 

 D personalData    DS                  INZ
 D  name                         30
 D   firstName                   15    Overlay(name)  
 D   lastName                    15    Overlay(name:16)
 D  phone_No                     10
 D   area_Code                    3    Overlay(phone_No)
 D   local_No                     7    Overlay(phone_No:*Next)
 D  status                        1    INZ(’I’)

 D myDataStruct    DS
 D  numField                      7s 0 
 D  charField                     7a   Overlay(myDataStruct)

Note the use of "floating" subfield names to show structure. 

Area_Code overlays first 3 positions of phone number (pos 31-33 of DS). Local_No overlays the last 
7 positions of phone number (pos 34-40 of DS). Use of OVERLAY to "subdivide" or redefine fields is 
recommended because of the increased maintainability.  If fields are added to the DS, or the length 
of a field is changed, fields defined in this way will "self-adjust" - unlike the old From - To method that 
requires modification whenever such changes are made.

The INZ keyword can be used at the DS level and/or the subfield level. Note the entire DS initialized 
to blanks except status, initialized to "I".

Rules for OVERLAY use:
Optional position parameter may be a numeric literal, numeric named constant, or a built-in function 
As of V4R4, *Next may be used instead of the fixed position parameter.
Length notation must be used on the overlaying field.
If an array is overlaid, keyword applies to each element of the array.

LIKE and DTAARA Keywords

Alternatives to DEFINE op code (previously DEFN)

LIKE keyword functions similarly to *LIKE DEFINE
Length, decimal positions and data type are copied

Length adjustments with + or - in To/Len column

May be used with fields, subfields, and arrays
 Arrays must still specify DIM keyword

i.e. If you reference an array with LIKE ... the DIM is not inherited 

  

  // Define a field LIKE an existing one
 D MonthTotal      S              7  2 
 D TotMonths       S             +2    LIKE(MonthTotal)
 D MonthArray      S                   LIKE(MonthTotal) DIM(12)

  // Define a field LIKE an array element
 D CompanyArray    S              3a   DIM(20)
 D CompanyCode     S                   LIKE(CompanyArray) 
 D StoreCode       S                   LIKE(CompanyCode) DTAARA(*LDA)
  // StoreCode is stored in the first 3 characters of the LDA

Although the dimensions of an array are not inherited along with the LIKE keyword, You can define 
one array exactly like another by using the built-in function %Elem to copy the number of elements in 
the parent array.

  D SaveRec       E DS                  ExtName(Product)  
  D                                     Prefix(’SAV.’)   
   

Keywords Used for Data Structures

Keyword  Description

INZ{(constant{:*ExtDft or *User})}  Initialize to a value

OCCURS(number)  Number of occurrences in multiple occurrence DS

OVERLAY(name:{pos or *Next})  Redefines subfields in a DS 

EXTNAME(name{:fmt_name}{: 
fieldtype} ) Data Structure is externally described

EXTFLD(fldname)  External field name being renamed

PREFIX(prefix_name)     Prefix subfields in an externally described DS

QUALIFIED Subfields in DS are qualified by DS name 

LIKEDS(DS_name) Creates identical subfields as in DS_name 

LIKEREC( fmt_name { : fieldtype} ) Similar to EXTNAME but can be used within a DS

DIM(number) Creates a Data Structure Array

Note that the PREFIX keyword can be used on externally described data structures to rename fields 
as it does on the F spec.  This can be quite useful if the programmer needs to save the values of a 
record to compare them to the next record’s values, for example.

If the prefix value is enclosed in quotes, it can include a "." (period) - this allows a qualified name to 
be created.  This same technique can be used on F-specs.  In the example shown, if the record 
format named PRODUCT contained a field named PRODCD, then that field would be renamed as 
SAV.PRODCD.     

The *EXTDFT parameter is only allowed for externally described data structures. It initializes 
externally described data-structure subfields with the default values from the DFT keyword in the 
DDS. You can override the value specified in the DDS by coding INZ with or without a parameter on 
the subfield specification. 

Note that we will be discussing some of the 3 V5R2 keywords in more detail on later charts.

The LIKEREC keyword differs from EXTNAME in one significant aspect - it can only reference a 
record format used in the program.

Keywords Short Description

ALIGN Align Integer & Float on natural boundary 

BASED(ptr) The basing pointer (ptr) contains the location of 
the storage holding the value for the field

DATFMT(format) Specifies the date type for a D field

EXPORT
The storage for the field is allocated in this 
module, but may be used in other modules in this 
program

EXTPGM(name) The program defined by this prototype 

EXTPROC(name) The procedure defined by this prototype

IMPORT Storage for this item is allocated in another 
module. 

OPTIONS(parm_passing_options) Used in prototypes to determine how parameter is 
passed 

PACKEVEN Zero out the "unused" digit of an even packed 
field

PROCPTR Defines a pointer as a procedure pointer. Only 
allowed with data type ’*’ (pointer)

STATIC The data item is allocated in static storage so that 
it will retain it’s value between procedure calls

TIMFMT Specifies the time type for a T field

Other D Spec Keywords

     FDisplay   CF   E             WORKSTN INDDS(DspInd)       
     D DspInd          DS                                        
      * Response indicators              
     D    Exit                 3      3N                         
     D    Return              12     12N                         
      * Conditioning indicators      
     D    Error               31     31N                         
     D    StDateErr           32     32N                         
     D    EndDateErr          33     33N
 

     C                   If        StartDate < Today 
     C                   Eval      StDateErr = *On
     C                   EndIf
    

     C                   Eval      EndDateErr = EndDate < StartDate
     C                   Eval      Error = StDateErr or EndDateErr
     C                   ExFmt     TestRec  
                          

     C                   If        Exit or Return
 

Named Indicators and INDDS

Use real names for your display indicators !
F-spec keyword INDDS ties the external indicators to the named DS

An alternative approach to the indicator defintion is shown on the next page 

This example illustrates the use of the INDDS (indicator Data Structure).

Note the definition of the DSPIND data structure.  When this is specified on the F spec, the 
programmer MUST use the indicators in the data structure and NOT the numbered indicators to 
control this file.   For example, in this program, if the programmer turned on indicator *IN31 directly 
in the program logic, it would have zero impact on the display file. The program logic must refer to 
the indicator as "Error" to turn it on or off in order for it to have an impact on the display file.

This is the DDS for the display file used in the RPG example:

                                         INDARA
  R          R TESTREC                   CF03(03) CF12(12)
                                     5 11’Start Date: . .’ 
               STARTDATE       L     5 27DATFMT(*USA)
     32                                  DSPATR(RI PC)
     32                              5 39’Date cannot be earlier than today’
                                     7 11’End Date: . . .’
               ENDDATE         L     7 27DATFMT(*USA)
     33                                  DSPATR(RI PC)
     33                              7 39’Date must be later than Start Date’
                                                                               
     31                             10 21’Please correct above error’  

                                                                               

In the example on the following page, we have also incorporated the associated indicator number in 
the indicator’s name.  Some people like this approach as it enables the programmer to see the 
indicator number without having to look at the Indicator Data Structure

     FDisplay   CF   E             WORKSTN INDDS(DspInd)       
     D DspInd          DS                                        
      * Response indicators              
     D    Exit_3                       N   Overlay(DspInd: 3)                 
     D    Return_12                    N   Overlay(DspInd: 12)                
      * Conditioning indicators      
     D    Error_31                     N   Overlay(DspInd: 31)                
     D    StDateErr_32                 N   Overlay(DspInd: 32)                
     D    EndDateErr_33                N   Overlay(DspInd: 33) 
      /Free
        If StartDate < Today;
          StDateErr_32 = *On;
        EndIf;
        EndDateErr_33 = EndDate < StartDate;
        Error_31 = StDateErr_32 or EndDateErr_33;
        ExFmt TestRec;  
                          

        If Exit_3 or Return_12;
 

Named Indicators and INDDS

Use the OVERLAY keyword to avoid From/To notation
Some people find this a more better approach

Note: In this case we have incorporated the indicator number in the name 

Mapping the *IN Array

Don’t want to use the INDDS approach?
You can always map the standard indicator array with a "Based" DS

Do you use the *INKx Indicators?
A similar approach can be used 

Use %Addr(*INKA) to initialize the basing pointer
Don’t foget that there are only 24 indicators in the set!

There is no *INKO 
See the Notes page for an example of how this is done 

 D DspInd          DS                  Based(pIndicators)   
   // Response indicators              
 D    Exit                         N   Overlay(DspInd: 3)   
 D    Return                       N   Overlay(DspInd: 12)  
   // Conditioning indicators      
 D    Error                        N   Overlay(DspInd: 31)  
 D    StDateErr                    N   Overlay(DspInd: 32)  
 D    EndDateErr                   N   Overlay(DspInd: 33)
                                                            
 D pIndicators     S               *   Inz(%Addr(*In))

Unlike the INDDS approach, these named indicators DO directly affect the content of their 
corresponding *IN indicator.  If we EVAL Error = *On, then indicator *IN30 was just turned on.  This 
often makes this a better approach for those who use program described (i.e. O-spec) based files 
rather than externally described printer files.

Those of you who use the *INKx series of indicators to identify function key usage need not feel left 
out.  A similar technique can be used.  IN this case the pointer is set to the address of *INKA.  The 
other 23 function key indicators are in 23 bytes that follow.  IBM have confirmed many times that for 
RPG IV this will always be the case.

     D FunctionKeys    DS                  Based(pKxIndicators)
     D   thisIsKC                      N   Overlay(FunctionKeys: 3)
     D   thisISKL                      N   Overlay(FunctionKeys: 12)

     D pKxIndicators   S               *   Inz(%Addr(*InKA))

